



Uji Normalitas *Kolmogorov-Smirnov* Menggunakan Microsoft Excel

Muhammad Ardhy Bisma¹✉, Ekra Sanggala¹

⁽¹⁾Prodi D4 Logistik Bisnis, Sekolah Vokasi, Universitas Logistik dan Bisnis Internasional, Bandung

DOI: 10.31004/jutin.v8i3.47332

✉ Corresponding author:

[bisma@ulbi.ac.id]

Article Info

Abstrak

Kata kunci:

Distribusi Normal;

Uji Normalitas;

Kolmogorov-Smirnov;

Microsoft Excel;

Fungsi Statistika

Satu hal penting dalam melakukan analisis statistik adalah uji normalitas. Tujuan dari uji normalitas adalah untuk menentukan apakah data yang akan digunakan dalam penelitian mengikuti distribusi normal atau tidak mengikuti distribusi normal. Salah satu uji normalitas adalah Uji Normalitas *Kolmogorov-Smirnov*. *Microsoft Excel* dilengkapi dengan *Statistical Function*. Dengan demikian Uji Normalitas *Kolmogorov-Smirnov* dapat dilakukan dengan bantuan *Microsoft Excel*. Sebuah contoh kasus berupa data massa banteng dengan ukuran *sample* 90 telah melewati uji normalitas dengan metode *Kolmogorov-Smirnov*, hasilnya data tersebut berdistribusi normal.

Abstract

Keywords:

Normal Distribution;

Normality Test;

Kolmogorov-Smirnov;

Microsoft Excel;

Statistical Function

One important thing in conducting statistical analysis is the normality test. The purpose of the normality test is to determine whether the data to be used in the research follows a normal distribution or does not follow a normal distribution. One of the normality tests is the Kolmogorov-Smirnov Normality Test. Microsoft Excel is equipped with Statistical Function. Thus the Kolmogorov-Smirnov Normality Test can be done with the help of Microsoft Excel. An example of a case in the form of bull mass data with a sample size of 90 has passed the normality test using the Kolmogorov-Smirnov method, the results of which are that the data is normally distributed.

1. PENDAHULUAN

Distribusi normal, juga dikenal sebagai distribusi *Gauss*, merupakan jenis distribusi variabel acak kontinyu yang digambarkan dengan kurva berbentuk lonceng simetris yang melandai pada bagian tepi dan memuncak pada bagian tengah (Johnson & Bhattacharyya, 2019). Distribusi normal ini digunakan untuk menggambarkan pola penyebaran data dari variabel kontinyu seperti tinggi badan, berat badan, nilai IQ, suhu udara dan lain lain (Ritonga et al., 2025).


Received 6 June 2025; Received in revised form 12 June 2025 year; Accepted 22 June 2025

Available online 5 July 2025 / © 2025 The Authors. Published by Jurnal Teknik Industri Terintegrasi Universitas Pahlawan Tuanku Tambusai. This is an open access article under the CC BY-SA license (<https://creativecommons.org/licenses/by-sa/4.0>)

Distribusi normal mempunyai *Probability Density Function (PDF)* sebagai berikut ini (Al-Oklah et al., 2014):

$$f(x) = \frac{1}{\sigma\sqrt{2\pi}} e^{-\frac{1}{2}\left(\frac{x-\mu}{\sigma}\right)^2}, -\infty < x < \infty$$


Jika persamaan *PDF* tersebut digambarkan dalam grafik, maka akan tergambar sebuah kurva yang menyerupai lonceng (Moshayedi, 2022).



Gambar 1: Grafik Probability Density Function Dari Distribusi Normal (Walpole et al., 1993)


Luas permukaan dibawah kurva menyatakan probabilitas dari *random variable* X, dimana nilai X berada diantara $x=x_1$ dan $x=x_2$. Luas permukaan dibawah kurva ini merupakan hasil integral dari *Probability Density Function* (Moshayedi, 2022).

$$P(x_1 < X < x_2) = \frac{1}{\sigma\sqrt{2\pi}} \int_{x_1}^{x_2} e^{-\frac{1}{2}\left(\frac{x-\mu}{\sigma}\right)^2} dx$$



Gambar 2: Luas Permukaan Dibawah Kurva Distribusi Normal (Walpole et al., 1993)

Jika nilai *random variable* X berada diantara $x=-\infty$ dan $x=x_2$ maka luas permukaan dibawah kurvanya disebut dengan *Cumulative Distribution Function (CDF)* (D'Agostino, 1986).



Gambar 3: Grafik Cumulative Distribution Function Dari Distribusi Normal

Untuk melakukan penelitian, analisis data merupakan hal penting untuk menentukan validitas dan keakuratan kesimpulan yang diambil. Tanpa adanya analisis yang baik, hasil penelitian bisa menjadi tidak tepat. Satu hal penting dalam melakukan analisis statistik adalah uji normalitas (Nurhaswinda et al., 2025). Tujuan dari uji normalitas adalah untuk menentukan apakah data yang akan digunakan dalam penelitian mengikuti distribusi normal atau tidak mengikuti distribusi normal (Nasrum, 2018).

Uji normalitas adalah langkah-langkah yang harus dilalui untuk mengetahui apakah suatu data berasal dari populasi yang berdistribusi normal atau berada dalam sebaran normal (Nuryadi et al., 2017). Contoh beberapa metode uji normalitas adalah Uji Normalitas *Kolmogorov-Smirnov*, Uji Normalitas *Chi-Square*, Uji Normalitas *Lilliefors*, Uji Normalitas *Shapiro-Wilk*, Uji Normalitas *Jarque-Berre*, Uji Normalitas *D'Agostino-Pearson* dan Uji Normalitas *Anderson-Darling* (Zaiontz, 2025).

Cara kerja Uji Normalitas *Kolmogorov-Smirnov* adalah dengan membandingkan nilai-nilai *Cumulative Distribution Function* dari data yang akan digunakan dengan nilai-nilai *Cumulative Distribution Function* dari distribusi normal. Jika selisih nilai-nilai tersebut tidak melebihi batas nilai yang sudah ditentukan maka dapat disimpulkan bahwa data yang akan digunakan berdistribusi normal, tetapi jika selisih nilai-nilai tersebut melebihi batas nilai yang sudah ditentukan maka disimpulkan bahwa data yang akan digunakan tidak berdistribusi normal (Fitri et al., 2023).

Microsoft Excel merupakan salah satu *software* yang paling penting dan terkenal yang digunakan oleh banyak perusahaan dan perorangan dalam menyelesaikan tugas-tugas harian mereka (Hossain, 2021). *Microsoft Excel* dilengkapi berbagai *function* untuk mempermudah melakukan berbagai perhitungan. Diantara berbagai *function* tersebut, tersedia banyak *function* yang dapat membantu melakukan berbagai perhitungan statistika (Triola, 2018).

Dengan adanya uji normalitas yang perlu dilakukan, adanya Uji Normalitas *Kolmogorov-Smirnov* dan adanya kemampuan *Microsoft Excel* dalam melakukan berbagai perhitungan statistika, maka pada *paper* ini akan dibahas mengenai Uji Normalitas *Kolmogorov-Smirnov* menggunakan *Microsoft Excel*.

2. METODE

A. Uji Normalitas *Kolmogorov-Smirnov*

Berikut ini merupakan langkah-langkah dalam Uji Normalitas *Kolmogorov-Smirnov* (Ibarra, 2024):

1. Membuat distribusi frekuensi dari data yang diuji.
2. Menetapkan bahwa nilai Frekuensi Relatif Kumulatif merupakan nilai *Cumulative Distribution Function* berdasarkan data yang diuji untuk setiap kelas pada distribusi frekuensi.
3. Menghitung nilai *Cumulative Distribution Function* berdasarkan distribusi normal untuk setiap kelas pada distribusi frekuensi. Nilai *Cumulative Distribution Function* ini dihitung menggunakan nilai tertinggi yang diijinkan dalam setiap kelas.
4. Menghitung nilai D untuk setiap kelas pada distribusi frekuensi, yaitu selisih antara *Cumulative Distribution Function* berdasarkan data yang diuji dengan *Cumulative Distribution Function* berdasarkan distribusi normal.
5. Menentukan nilai D *maximal*, yaitu nilai D tertinggi dari seluruh nilai D untuk setiap kelas.
6. Menentukan nilai D kritis, yaitu nilai D tertinggi yang diijinkan. Nilai D kritis ini dapat diperoleh dari tabel *Kolmogorov-Smirnov Test Critical Values*. Tabel ini dapat dilihat pada lampiran.
7. Jika nilai D *maximal* tidak melebihi nilai D kritis, maka data berdistribusi normal, sedangkan jika nilai D *maximal* melebihi nilai D kritis, maka data tidak berdistribusi normal.

B. Statistical Function in Excel

Berikut ini *Statistical Function* yang akan digunakan (David, 2017):

- COUNT, untuk menghitung banyaknya data.
- MAX, untuk mencari nilai tertinggi dari data.
- MIN, untuk mencari nilai terendah dari data.
- COUNTIFS, untuk menghitung banyaknya data berdasarkan beberapa kriteria.
- NORM.DIST, untuk menghitung nilai *Cumulative Distribution Function* dari distribusi normal.

3. HASIL DAN PEMBAHASAN

Sebagai contoh penyelesaian Uji Normalitas *Kolmogorov-Smirnov* Menggunakan *Microsoft Excel* akan digunakan data sebagai berikut ini (Ibarra, 2024):

	A	B	C	D	E	F	G	H	I
1	Data								
2	Massa Banteng (Kg)								
3									
4	500,5	495,2	506,3	502,7	502,5	503,6	480,2	511,1	519,5
5	508,2	491,7	504,8	502,8	495,1	510,9	529,9	489,7	501,1
6	519,1	507,6	500,8	516,2	508,2	473,9	506,6	521,1	494,2
7	480,3	510,3	505,9	519,2	517,5	490,5	488,9	502,7	518,6
8	488,8	500,8	522,3	485,9	507,1	495,2	468,3	488,3	488,2
9	498,7	503,4	467,2	491,7	524,0	506,7	483,9	514,4	481,9
10	507,2	512,1	501,0	505,0	494,0	494,3	478,2	493,0	500,7
11	517,5	465,4	525,5	486,3	477,4	518,6	508,9	495,4	485,4
12	501,9	502,4	503,4	519,0	497,8	519,7	508,2	481,3	501,8
13	499,5	505,5	503,5	500,0	496,9	529,1	529,8	517,3	485,7
14									
15	μ (Kg)	501,37							
16	σ (Kg)	14,39							

Gambar 4: Data yang Akan Digunakan Sebagai Contoh Uji Normalitas Kolmogorov-Smirnov

Hipotesis yang akan digunakan adalah sebagai berikut ini (Ibarra, 2024):

- H0 : Data massa banteng berdistribusi normal dengan $\mu=501,37$ Kg dan $\sigma=14,39$ Kg.
H1 : Data massa banteng tidak berdistribusi normal dengan $\mu=501,37$ Kg dan $\sigma=14,39$ Kg.

Berikut ini merupakan contoh penyelesaian Uji Normalitas *Kolmogorov-Smirnov* dengan *Microsoft Excel* berdasarkan langkah-langkah pada bagian *methods*.

1. Membuat Distribusi Frekuensi.

- a) Satuan Terkecil
Berdasarkan data dapat diketahui bahwa satuan terkecil adalah 0,1.
- b) Jumlah Data
=COUNT(A4:I13) maka diperoleh 90.
- c) Nilai Maks (Kg)
=MAX(A4:I13) maka diperoleh 529,9.
- d) Nilai Min (Kg)
=MIN(A4:I13) maka diperoleh 465,4.
- e) Rentang (Kg)
=B24-B25 maka diperoleh 64,5.
- f) Banyak Kelas Interval
Untuk menghitung banyak kelas interval dapat digunakan persamaan berikut ini (Shayib, 2018):

$$k = 1 + 3,3 \log n, \text{ where } n \text{ is sample size}$$

=ROUND(1+(3,3*LOG(B23));0) maka diperoleh 7.

Maka tabel distribusi frekuensinya akan menjadi seperti berikut ini:

Tabel 1: Tabel Distribusi Frekuensi

	A	B	C	D	E	F	G	H	I	J	K	L
	Massa Banteng (Kg)											
31	Kelas	X	Batas Bawah (Kg)	Batas Atas (Kg)	Titik Tengah (Kg) (Xi)	Frekuensi (Fi)	Frekuensi Relatif	Frek. Rel. Kum	CDF	D		
32	1	-										
33	2	-										
34	3	-										
35	4	-										
36	5	-										
37	6	-										
38	7	-										
39						Jumlah						

- g) Panjang Kelas (Kg)
=ROUND(B26/B27;1) maka diperoleh 9,2.
- h) Nilai Terendah Yang Dijinkan Dalam Setiap Kelas (Kg)
=IF(A32=1;\$B\$25;D31+\$B\$22) maka untuk kelas 1 diperoleh 465,4. Copy-kan formula tersebut untuk kelas 1 sampai dengan 7.
- i) Nilai Tertinggi Yang Dijinkan Dalam Setiap Kelas (Kg)

- =**B32+\$B\$28** maka untuk kelas 1 diperoleh 474,6. Copy-kan formula tersebut untuk kelas 1 sampai dengan 7.
- j) Batas Bawah (Kg)
=B32-(\$B\$22/2) maka untuk kelas 1 diperoleh 465,35. Copy-kan formula tersebut untuk kelas 1 sampai dengan 7.
- k) Batas Atas (Kg)
=D32+(\$B\$22/2) maka untuk kelas 1 diperoleh 474,65. Copy-kan formula tersebut untuk kelas 1 sampai dengan 7.
- l) Titik Tengah (Kg)
=(B32+D32)/2 maka untuk kelas 1 diperoleh 470,0. Copy-kan formula tersebut untuk kelas 1 sampai dengan 7.
- m) Frekuensi
=COUNTIFS(\$A\$4:\$I\$13;">="&B32;\$A\$4:\$I\$13;"<="&D32) maka untuk kelas 1 diperoleh 4. Copy-kan formula tersebut untuk kelas 1 sampai dengan 7.
- n) Jumlah Frekuensi
=SUM(H32:H38) maka diperoleh 90.
- o) Frekuensi Relatif
=H32/\$H\$39 maka untuk kelas 1 diperoleh 0,044444444. Copy-kan formula tersebut untuk kelas 1 sampai dengan 7.
- p) Jumlah Frekuensi Relatif
=SUM(I32:I38) maka diperoleh 1,00.
- q) Frekuensi Relatif Kumulatif
=IF(A32=1;I32;I32+J31) maka untuk kelas 1 diperoleh 0,044444444. Copy-kan formula tersebut untuk kelas 1 sampai dengan 7. Maka distribusi frekuensinya seperti berikut ini:

Tabel 2: Distribusi Frekuensi

A	B	C	D	E	F	G	H	I	J
	Massa Banteng (Kg)								
Kelas	X			Batas Bawah (Kg)	Batas Atas (Kg)	Titik Tengah (Kg) (Xi)	Frekuensi (Fi)	Frekuensi Relatif	Frek. Rel. Kum
1	465,4	-	474,6	465,35	474,65	470,0	4	0,044444444	0,044444444
2	474,7	-	483,9	474,65	483,95	479,3	7	0,077777778	0,122222222
3	484,0	-	493,2	483,95	493,25	488,6	13	0,144444444	0,266666667
4	493,3	-	502,5	493,25	502,55	497,9	22	0,244444444	0,511111111
5	502,6	-	511,8	502,55	511,85	507,2	24	0,266666667	0,777777778
6	511,9	-	521,1	511,85	521,15	516,5	14	0,155555556	0,933333333
7	521,2	-	530,4	521,15	530,45	525,8	6	0,066666667	
						Jumlah	90	1,00	1

- Menetapkan Nilai Frekuensi Relatif Kumulatif Sebagai Nilai *Cumulative Distribution Function*.
Nilai Frekuensi Relatif Kumulatif dalam setiap kelas merupakan nilai *Cumulative Distribution Function* berdasarkan data yang diuji.
- Menghitung Nilai *Cumulative Distribution Function*.
=NORM.DIST(D32;\$B\$15;\$B\$16;1) maka untuk kelas 1 diperoleh 0,03142. Copy-kan formula tersebut untuk kelas 1 sampai dengan 7.
- Menghitung Nilai D.
=ABS(J32-K32) maka untuk kelas 1 diperoleh 0,013024288. Copy-kan formula tersebut untuk kelas 1 sampai dengan 7.
- Menentukan Nilai D *Maximal*.
=MAX(L32:L38) maka diperoleh 0,021828292. Copy-kan formula tersebut untuk kelas 1 sampai dengan 7.
- Menentukan Nilai D Kritis.
Untuk contoh ini nilai α yang digunakan adalah 0,05, maka berdasarkan tabel *Kolmogorov-Smirnov Test Critical Values*, nilai D Kritis adalah 0,141.
- Kesimpulan.

=IF(C43<=C42;"H0 Diterima";"H0 Ditolak") maka diperoleh "H0 Diterima".


Dengan demikian secara lengkap, tabel distribusi frekuensinya adalah sebagai berikut ini:

Tabel 3: Distribusi Frekuensi dan Hasil Uji Normalitas Kolmogorov-Smirnov

A	B	C	D	E	F	G	H	I	J	K	L
29	Massa Banteng (Kg)										
30	Kelas	X		Batas Bawah (Kg)	Batas Atas (Kg)	Titik Tengah (Kg) (Xi)	Frekuensi (Fi)	Frekuensi Relatif	Frek. Rel. Kum	CDF	D
32	1	465,4	-	474,6	465,35	474,65	470,0	4	0,044444444	0,044444444	0,03142 0,013024288
33	2	474,7	-	483,9	474,65	483,95	479,3	7	0,077777778	0,122222222	0,11237 0,009855525
34	3	484,0	-	493,2	483,95	493,25	488,6	13	0,144444444	0,266666667	0,28510 0,018433872
35	4	493,3	-	502,5	493,25	502,55	497,9	22	0,244444444	0,511111111	0,53130 0,020184365
36	5	502,6	-	511,8	502,55	511,85	507,2	24	0,266666667	0,777777778	0,76572 0,012062417
37	6	511,9	-	521,1	511,85	521,15	516,5	14	0,155555556	0,933333333	0,91483 0,018506623
38	7	521,2	-	530,4	521,15	530,45	525,8	6	0,066666667		1 0,97817 0,021828292
39							Jumlah	90	1,00		
40											
41	α	0,05									
42	D Kritis	0,141									
43	D Max	0,021828292									
44	Kesimpulan	H0 Diterima									

Diketahui bahwa nilai D *Maximal* adalah 0,021828292 dan nilai D *Kritis* adalah 0,141, maka nilai D *Maximal* tidak melebihi nilai D *Kritis*, dengan demikian dapat disimpulkan bahwa H0 diterima atau data berdistribusi normal dengan $\mu=501,37$ Kg dan $\sigma=14,39$ Kg.

Nilai Frekuensi Relatif Kumulatif dan *Cumulative Distribution Function* jika ditampilkan dalam grafik akan seperti berikut ini:



Gambar 5: Grafik Frekuensi Relatif Kumulatif dan Cumulative Distribution Function

Dari grafik diatas terlihat bahwa kurva Frekuensi Relatif Kumulatif nyaris berhimpit dengan kurva *Cumulative Distribution Function*, sehingga secara kasat mata pun dapat disimpulkan bahwa data berdistribusi normal.

4. KESIMPULAN

Uji Normalitas *Kolmogorov-Smirnov* merupakan uji normalitas yang cukup mudah untuk dilakukan karena langkah-langkah dan perhitungannya sangat mudah. Untuk melakukan Uji Normalitas *Kolmogorov-Smirnov* tidak perlu dilakukan dengan *software* statistika yang populer seperti SPSS, R dan SAS, tetapi juga dapat dilakukan dengan *software* yang sangat familiar digunakan dalam tugas sehari-hari, yaitu *Microsoft Excel*. *Microsoft Excel* ini, selain dilengkapi oleh *function* yang sangat umum digunakan setiap hari seperti SUM, COUNT, INDEX, MATCH dan lainnya, dilengkapi juga dengan *statistical function* seperti NORM.DIST, POISSON.DIST, EXPON.DIST, AVERAGE, STDEV.P dan lain-lain.

Keunggulan dari penggunaan *Microsoft Excel* dalam melakukan Uji Normalitas *Kolmogorov-Smirnov* adalah dapat membantu dalam memahami langkah-langkahnya, karena setiap perhitungan dilakukan secara mandiri memanfaatkan *statistical function* yang sudah disediakan. Hal ini tentunya sangat berbeda dengan *software* statistika seperti SPSS, R dan SAS, dimana telah menyediakan *feature* khusus untuk melakukan Uji Normalitas *Kolmogorov-Smirnov* secara otomatis dari awal hingga selesai.

Kelemahan dari penggunaan *Microsoft Excel* dalam melakukan Uji Normalitas *Kolmogorov-Smirnov* adalah jika ukuran data yang akan diuji sangat besar, maka perhitungannya akan sangat lama. Hal ini sangat berbeda dengan *software* statistika seperti SPSS, R dan SAS yang mempunyai kemampuan lebih baik untuk menangani ukuran data yang sangat besar.

5. REFERENSI

- Al-Oqlah, H., Titi, S., & Alodat, T. (2014). *Introduction to statistics made easy*. King Saud University Pres.
- D'Agostino, R. B. (1986). *Goodness-of-fit-techniques* (Vol. 68). CRC press.
- David, M. (2017). *Statistics for managers, using Microsoft excel*. Pearson Education India.
- Fitri, A., Rahim, R., Nurhayati, N., Azis, A., Pagiling, S. L., Natsir, I., Munfarikhatin, A., Simanjuntak, D. N., Hutagaol, K., & Anugrah, N. E. (2023). *Dasar-dasar Statistika untuk Penelitian*.
- Hossain, E. (2021). *Excel Crash Course for Engineers*. Springer.
- Ibarra, L. F. (2024). *Simulation and Statistics with Excel: An Introduction to Business Students*. CRC Press.
- Johnson, R. A., & Bhattacharyya, G. K. (2019). *Statistics: principles and methods*. John Wiley & Sons.
- Moshayedi, N. (2022). *Introduction to Probability Theory: A First Course on the Measure-Theoretic Approach*. World Scientific.
- Nasrum, A. (2018). Uji normalitas data untuk penelitian. *Jayapangus Press Books*, i–117.
- Nurhaswinda, N., Zulkifli, A., Gusniati, J., Zulefni, M. S., Afendi, R. A., Asni, W., & Fitriani, Y. (2025). Tutorial uji normalitas dan uji homogenitas dengan menggunakan aplikasi SPSS. *Jurnal Cahaya Nusantara*, 1(2), 55–68.
- Nuryadi, Astuti, T. D., Utami, E. S., & Budiantara, M. (2017). Buku Ajar Dasar-dasar Statistik Penelitian. In *Sibuku Media*.
- Ritonga, A., Saragih, E. L., Purba, G. A., Pandiangan, P. P. S., Damanik, R. N., & Al Azmi, F. (2025). Penerapan Distribusi Normal Dalam Pengukuran Tinggi Badan Mahasiswa FMIPA Universitas Negeri Medan 2024. *Bilangan: Jurnal Ilmiah Matematika, Kebumian Dan Angkasa*, 3(2), 39–53.
- Shayib, M. A. (2018). *Descriptive Statistics: The Basic for Biostatistic Volume I*. USA: Bookboon, The EBook Company.
- Triola, M. F. (2018). *Elementary statistics using Excel*. Pearson.
- Walpole, R. E., Myers, R. H., Myers, S. L., & Ye, K. (1993). *Probability and statistics for engineers and scientists* (Vol. 5). Macmillan New York.
- Zaiontz, C. (2025). *Statistical Tests for Normality and Symmetry.pdf*. <https://real-statistics.com/tests-normality-and-symmetry/statistical-tests-normality-symmetry/>